
I. Berkeley Madonna script used to solve equations 8-10

;donor plate
d/dt(Dd[1..N])  = -kon*Da[i]*Dd[i] +koff*Dda[i]  +Pe*A*(Md[i]/Vm)  -Pe*A*(Dd[i]/Vd)
d/dt(Da[1..N]) = -kon*Da[i]*Dd[i] +koff*Dda[i]                         
d/dt(Dda[1..N])=  kon*Da[i]*Dd[i] -koff*Dda[i]

;membrane
d/dt(Md[1..N]) =     -Pe*A*(Md[i]/Vm)  +Pe*A*(Dd[i]/Vd) -Pe*A*(Md[i]/Vm) +Pe*A*(Ad[i]/Vd)

;acceptor plate
d/dt(Ad[1..N])  =           +Pe*A*(Md[i]/Vm) -Pe*A*(Ad[i]/Vd)

init(Dd[1..N]) = 50*10^-6       ;free drug in donor plate, M
init(Da[1]) = 0                    ;protein in Donor plate, M
init(Da[2]) = 100*10^-6            ;protein in Donor plate, M
init(Dda[1..N]) = 0                ;complex drug-albumin in Donor, M 
init(Ad[1..N]) = 0             ;drug in acceptor plate, M 
init(Md[1..N]) = 0             ;drug in membrane, M 

DTD[1..N]=Dd[i]+Dda[i]              ;total drug in Donor, M
ATD[1..N]=Ad[i]                     ;total drug in Acceptor, M
ATD1=Ad[1]           
ATD2=Ad[2]

N=2  ;number of experiments in the array
Pe = 1*10^-3    ;permeability, cm/s
A = 0.048        ;area(cm2)*porosity
Vr = 0.3           ;acceptor volume, mL
Vd = 0.3           ;donor volume, mL
Vm = 7.5e-4*Kh     ;volume membrane, mL, multiplied by partition coefficent to give apparent volume
Kh = 100           ;hexadecane partition coefficient
kon = koff/Kd
koff = 1 ;s-1 
Kd = 1.37*10^-4  ;M

METHOD STIFF        ;integration method
STARTTIME = 0
STOPTIME = 7200      ;seconds
TOLERANCE = 0.001    ;relative accuracy for the auto and stiff methods
DT = 0.001                   ;Integration time step (Euler, RK2, RK4 methods)
DTMIN = 0.001
DTMAX = 1
DTOUT = 0                   ;output time interval (0 = store every step)

II. Details of the PAMPA construct:

Acceptor plate 

Filterplate reference used with hexadecane membrane:

Millipore, Multiscreen plate, ref MPC4NTR10

Filterplate reference used with octanol membrane:

Millipore, Multiscreen plate, ref S2EJ014L01

Donor plate (Teflon, non-disposable):

Millipore reference: MSSACCEPTOR


